Modeling Sensors and other Physical Systems with SPICE

(Part 2)

Dr. José Ernesto Rayas-Sánchez

1

Characterization of Sensors and Transducers

- Analytical functions
  - Multidimensional vector functions √
  - Nonlinear systems of equations
  - Systems of nonlinear differential equations
- Measurements

Dr. J. E. Rayas-Sánchez

2

System of Nonlinear Equations



$\boldsymbol{x}$ Vector of physical variables (temperature, humidity, pressure, volumetric flow, etc.)

$\boldsymbol{y}$ Vector of electrical outputs (voltage, current, charge, resistance, etc.)

$$\boldsymbol{f}(\boldsymbol{x}, \boldsymbol{y}) = \boldsymbol{0}$$

$$\boldsymbol{x} \in \Re^n \qquad \boldsymbol{f}, \boldsymbol{y} \in \Re^m \qquad \boldsymbol{f} : \Re^{n+m} \rightarrow \Re^m$$

Implementing $\boldsymbol{f}(\boldsymbol{x}, \boldsymbol{y}) = \boldsymbol{0}$ with SPICE

1) Implement each physical variable, x_i, …, x_n with a DC voltage source, isolated from the rest of the circuit
2) Implement each constant or any other parameter using also an isolated DC voltage source
3) Use a linear or nonlinear voltage controlled current source to implement each function f_1, …, f_m, and connect them to the following infinite input impedance subcircuit named `Variable`

Implementing $\boldsymbol{f}(\boldsymbol{x}, \boldsymbol{y}) = \mathbf{0}$ with SPICE (cont.)



SPICE finds the voltage y_i such that the current $f_i = 0$, for $i = 1 \ldots m$

Subcircuit `Variable`




```
.SUBCKT Variable Yi
RR Yi internal 1
VV internal 0 0V
FIv 0 Yi VV 1
.ENDS
```

Example of Simulating $\boldsymbol{f}(\boldsymbol{x}, \boldsymbol{y}) = \boldsymbol{0}$

$$\boldsymbol{f}(\boldsymbol{x}, \boldsymbol{y}) = \begin{bmatrix} 10y_1 - x_1 \\ y_1 + x_2 y_2 \end{bmatrix} = \boldsymbol{0}$$

$$x_1 = [0,10]$$

$$x_2 = 0.5$$

```
System of Equations, Example 1

Vx1 x1 0 DC 1
Vx2 x2 0 DC 0.5
b1 0 y1 i = 10*v(y1) - v(x1)
XY1 y1 Variable
b2 0 y2 i = v(y1) + v(x2)*v(y2)
XY2 y2 Variable
.NODESET V(y1)=0 V(y2)=0

.SUBCKT Variable Yi
RR Yi internal 1
VV internal 0 0V
FIv 0 Yi VV 1
.ENDS

.control
DC Vx1 0 10 1
plot v(y1) v(y2)
.endc
.end
```


Example of Simulating $\boldsymbol{f}(\boldsymbol{x}, \boldsymbol{y}) = \boldsymbol{0}$ (cont.)

$$\boldsymbol{f}(\boldsymbol{x}, \boldsymbol{y}) = \begin{bmatrix} 10y_1 - x_1 \\ y_1 + x_2 y_2 \end{bmatrix} = \boldsymbol{0}$$

$$x_1 = [0,10]$$

$$x_2 = 0.5$$



Characterization of Sensors and Transducers

- Analytical functions
  - Multidimensional vector functions √
  - Nonlinear systems of equations √
  - Systems of nonlinear differential equations
- Measurements

System of Nonlinear Differential Equations



$\boldsymbol{x}$ Vector of physical variables (temperature, humidity, pressure, volumetric flow, etc.)

$\boldsymbol{y}$ Vector of electrical outputs (voltage, current, charge, resistance, etc.)

$$\boldsymbol{f}(\boldsymbol{x}, \boldsymbol{y}, \boldsymbol{y}', t) = \boldsymbol{0}$$

$$\boldsymbol{x} \in \Re^{n} \qquad \boldsymbol{f}, \boldsymbol{y} \in \Re^{m} \qquad \boldsymbol{y}' \in \Re^{m'} \qquad t \in \Re \qquad \boldsymbol{f} : \Re^{n+m+m'+1} \rightarrow \Re^{m}$$

y' contains the first order derivatives w.r.t time *t* of some of the elements of ***y***

Implementing $\boldsymbol{f}(\boldsymbol{x}, \boldsymbol{y}, \boldsymbol{y}', t) = \mathbf{0}$ with SPICE

1) Implement each physical variable, x_i, …, x_n with a DC voltage source, isolated from the rest of the circuit
2) Implement each constant or any other parameter using also an isolated DC voltage source
3) Use a voltage controlled current source to implement each function f_1, …, f_m,
4) If function f_i does not contain a derivative, connect f_i to the subcircuit named `Variable`
5) If function f_i contains a derivative, connect f_i to the following infinite input impedance subcircuit named `Derivative`

Implementing $\boldsymbol{f}(\boldsymbol{x}, \boldsymbol{y}, \boldsymbol{y}', t) = \mathbf{0}$ with SPICE (cont.)



SPICE finds the voltage y_i such that the current $f_i = 0$

$$i_V = C\frac{dy_i}{dt} = y_i'$$

Subcircuit `Derivative`




```
.SUBCKT Derivative Yi Yi_prime
CC Yi internal 1
VV internal 0 0V
FIv 0 Yi VV 1
HIv Yi_prime 0 VV 1
.ENDS
```


Ejemplo de Derivación con SPICE

$$x = \sin(2\pi t)$$

$$\frac{dx}{dt} = ?$$

```
Derivative, Example 1
Vx  in 0 DC 0 SIN(0 1 1)
Xx_dxdt in out Derivative
.SUBCKT Derivative Yi Yi_prime
CC Yi internal 1
VV internal 0 0V
FIv 0 Yi VV 1
HIv Yi_prime 0 VV 1
.ENDS
.control
destroy all
TRAN 1ms 3s
plot v(in) v(out)
.endc
.end
```

Ejemplo de Derivación con SPICE (cont.)

$x = \sin(2\pi t) \qquad \frac{dx}{dt} = ?$



Ejemplo de Solución de ***f*(*y*, *y'*, *t*) = 0**

$$\frac{dy}{dt} = 1 - y^2 \quad \rightarrow \quad f(y,t) = \frac{dy}{dt} - 1 + y^2 = 0$$

$$y(0) = 0$$



Ejemplo de Solución de ***f*(*y*, *y'*, *t*) = 0** (cont.)

$$f(y,t) = \frac{dy}{dt} - 1 + y^2 = 0$$

$$y(0) = 0$$

```
Solving SNL_Diff_Eq, Example 1
b1 0 in i = v(out) - 1 + v(in)*v(in)
Xy_dydt in out Derivative
.IC v(in) = 0
.SUBCKT Derivative Yi Yi_prime
CC Yi internal 1
VV internal 0 0V
FIv 0 Yi VV 1
HIv Yi_prime 0 VV 1
.ENDS
.control
TRAN 1ms 3s
plot v(in) v(out)
.endc
.end
```


Ejemplo de Solución de ***f*(*y*, *y'*, *t*) = 0** (cont.)

$$f(y,t) = \frac{dy}{dt} - 1 + y^2 = 0$$

$$y(0) = 0$$



Ejemplo de Solución de ***f*(*y*, *y'*, *t*) = 0** (cont.)

$$f(y,t) = \frac{dy}{dt} - 1 + y^2 = 0$$

$$y(0) = 0$$

Otra forma equivalente:

```
Solving SNL_Diff_Eq, Example 1
GG 0 in poly(2) out 0 in 0 -1 1 0 0 0 1
Xy_dydt in out Derivative
.IC v(in) = 0
.SUBCKT Derivative Yi Yi_prime
CC Yi internal 1
VV internal 0 0V
FIv 0 Yi VV 1
HIv Yi_prime 0 VV 1
.ENDS
.control
TRAN 1ms 3s
plot v(in) v(out)
.endc
.end
```